
PROGRAMA PARA COMPARACIONES PARTICULARES  DE LA PRUEBA 
D.E NO PROTEGIDA 
 
C PROGRAMA PRINCIPAL. 
C 
C ARCHIVO: PDENP.FOR 
C REVISION: 21 DE NOVIEMBRE DE 2003 
C 
C PROGRAMADO POR: PAOLA RUELAS VILLEGAS. 
C                 ESTEBAN BURGUETE. 
C                 JOSE FRANCISCO TAMBORERO. 
C  
C PROGRAMA PARA CALCULAR EL ERROR EN LAS COMPARACIONES  
C PARTICULARES DE LA PRUEBA D.E. NO PROTEGIDA EN EL  
C MODELO COMPLETAMENTE AL AZAR. 
C  
C LOS ERRORES SE ASUMEN INDEPENDIENTES Y NORMALMENTE  
C DISTRIBUIDOS CON MEDIA CERO Y VARIANZA COMÚN SIGMA  
C CUADRADA.  
C 
C SE HACE USO DEL METODO DE PLUG IN BOOTSTRAP O BOOTSTRAP 
C PARAMETRICO. ESTE METODO REQUIERE LA SIMULACION DEL MODELO 
C ORIGINAL REDUCIDO POR LA HIPOTESIS NULA, PERO CONSIDERANDO  
C LOS PARAMETROS RESTANTES. VER LA TESIS: 
C 
C        "ESTIMACION DE FWE, PCE Y PFE DE LA PRUEBA DE 
C         EN EL MODELO COMPLETAMENTE AL AZAR."  
C         DEPARTAMENTO DE INGENIERIA INDUSTRIAL.  
C         UNIVERSIDAD DE LAS AMERICAS-PUEBLA. 
C 
C EL METODO PLUG IN BOOTSTRAP HACE USO  
C DE LA SIMULACION. LA SIMULACION CONSIDERA DOS PARAMETROS 
C QUE SON: EL NUMERO EL NUMERO DE SIMULACIONES(IB) Y EL 
C NUMERO DE ITERACIONES POR SIMULACION(IREPB). EL NUMERO DE  
C ITERACIONES POR SIMULACION SIRVE PARA CALCULAR EL VALOR P 
C DE LAS COMPARACIONES, PARA ESTE PROGRAMA 
C IB REPRESENTA EL NUMERO DE VECES QUE SE SIMULA UN CONJUNTO DE  
C DATOS CON UN CIERTO NUMERO DE TRATAMIENTOS Y DE REPETICIONES. 
C MIENTRAS QUE EL  
C NUMERO DE SIMULACIONES SIRVE PARA COLECTAR NUMERO DE RECHAZOS  
C ESTIMADOS Y ESTABLECER EL VALOR P REAL DE CADA  
C UNA DE LAS COMPARACIONES. CABE MENCIONAR  
C QUE EL USUARIO PUEDE FIJAR ESTOS VALORES, LA RECOMENDACION 
C ES USAR COMO MINIMO 100000 SIMULACIONES Y 10000 ITERACIONES  
C POR SIMULACION.  
C 
C 
C COMO LINEAS DE PROGRAMACION, SE TIENEN: 
C 
C             1. EL LENGUAJE DE PROGRAMACION USADO ES EL 
C                FORTRAN 
C             2. LAS VARIABLES Y FUNCIONES SERAN ENTERAS SI 
C                PRINCIPIAN CON LAS LETRAS I,J,K,L,M,N. Y  
C                SERAN REALES SI PRINCIPIAN CON CUALQUIER  
C                OTRA LETRA. 
C             3. DEBIDO A PROBLEMAS CON EL RECONOCIMIENTO DE 
C                ALGUNOS CARACTERES, SE HA DECIDIDO NO USAR  
C                ACENTOS. POR LO MISMO, SE DECIDIO USAR  
C                MAYUSCULAS EN LOS COMENTARIOS. EN EL PROGRAMA 

C                PUEDEN APARECER LINEAS EN MINUSCULAS. 
C             4. EL PROGRAMA TRABAJA INTERACTIVAMENTE. AL  
C                PRINCIPIO PREGUNTA CUAL ES EL ARCHIVO 
C                DE SALIDA, ASI COMO LA SEMILLA  
C                PARA LA SIMULACION. 
C 
C SE DEFINEN ALGUNOS PARAMETROS 
C 
      CHARACTER*12 SALIDA 
      COMMON C0,C1,C2,D1,D2,D3 
      COMMON SALIDA 
      COMMON IB,IREPB 
C 
C A CONTINUACION SE ASIGNAN LAS CONSTANTES                 
C QUE SE UTILIZAN PARA GENERAR 
C LOS NUMEROS ALEATORIOS NORMALES CON MEDIA CERO Y  
C VARIANZA UNO.   
C  
C PARA MAYOR INFORMACION VER LA FORMULA  
C 26.1 DE: 
C 
C        ABAMOWITZ & STEGUN.  
C        A HANDBOOK OF MATHEMATICAL FUNCTIONS. 
C        NATIONAL BUREAU OF STANDARDS.  
C        WASHINGTON,DC. 
C 
      C0=2.515517       
      C1=0.802853 
      C2=0.010328 
      D1=1.432788 
      D2=0.189269 
      D3=0.001308 
C  
C AQUI SE LEE EL NOMBRE DEL ARCHIVO DE  
C SALIDA. 
C EL ARCHIVO DE SALIDA LO CONSTRUYE EL PROGRAMA. SI YA  
C EXISTE UN ARCHIVO CON ESE NOMBRE, POR PROTECCION, 
C EL PROGRAMA SE DETIENE CON UN MENSAJE DE ERROR.   
C 
    1 FORMAT(A12) 
      PRINT*,'  ' 
      PRINT*,'   INTRODUCE EL NOMBRE DEL ARCHIVO DE SALIDA(OUTPUT)' 
      READ(*,1)SALIDA 
      OPEN(5,FILE=SALIDA,STATUS='NEW') 
C       
C SE PONE EL MENSAJE INICIAL EN EL ARCHIVO DE SALIDA.       
C  
      WRITE(5,179) 
  179 FORMAT(/ 
     110X,'PROGRAMA PARA CALCULAR EL VALOR P DE LAS'/ 
     110X,'COMPARACIONES PARTICULARES NO PROTEGIDAS DE  '/ 
     210X,'LA PRUEBA D.E. EN EL DISEÑO COMPLETAMENTE AL AZAR,'// 
     510X,'PROGRAMADO POR:PAOLA RUELAS VILLEGAS.'/ 
     610X,'                    ESTEBAN BURGUETE.'/ 
     710X,'                      JOSE TAMBORERO.'/) 
C 
C SE MANDAN LOS NOMBRES DE LOS ARCHIVOS DE ENTRADA Y SALIDA AL 
C ARCHIVO DE SALIDA.         
C               
      WRITE(5,180)SALIDA 

  180 FORMAT(3(/),10X,'ARCHIVO DE  SALIDA: ',A12)     
C 
C SE PIDE QUE SE INTRODUZCA LA SEMILLA. 
C 
   34 PRINT*,' '      
      PRINT*,'   INTRODUCE LA SEMILLA:'  
      PRINT*,'   DEBE SER UN ENTERO MAYOR QUE  0' 
      PRINT*,'   Y MENOR QUE 2147483647' 
      READ*,ISEMILLA 
      IF(ISEMILLA.LE.0.OR.ISEMILLA.GE.2147483647)GO TO 34 
C 
C LA SEMILLA SE GRABA EN EL ARCHIVO DE SALIDA. 
C       
      WRITE(5,181)ISEMILLA 
  181 FORMAT(//10X,'SEMILLA PARA SIMULACION: ',I10) 
C 
C  
C POR SEGURIDAD, SE DESHECHAN LOS PRIMEROS 1000 NUMEROS  
C GENERADOS POR LA SEMILLA. ESTO SE HACE PARA ALEATORIZAR  
C MEJOR LA SEMILLA, YA QUE POR CADA NUMERO QUE SE GENERA, SE 
C OBTIENE UNA NUEVA SEMILLA. 
C 
      DO 5 II=1,1000 
      YY=UNIFO(ISEMILLA) 
    5 CONTINUE 
C 
C 
C SE FIJAN EL NUMERO DE SIMULACIONES(IB) Y EL NUMERO DE 
C ITERACIONES POR SIMULACION(IREPB). 
C                                            
      PRINT*,'  ' 
      PRINT*,'   INTRODUCE EL NUMERO DE SIMULACIONES Y' 
      PRINT*,'   EL DE ITERACIONES POR SIMULACION' 
      READ*,IB,IREPB      
C 
C ESTA INFORMACION SE MANDA AL ARCHIVO DE SALIDA. 
C 
      WRITE(5,188)IB,IREPB 
  188 FORMAT(//10X,'NUMERO TOTAL DE SIMULACIONES: ',I6/10X, 
     1            'ITERACIONES  POR  SIMULACION: ',I6) 
C 
C SE PIDE EL NIVEL DE CONFIANZA DESEADO PARA EL INTERVALO 
C DE CONFIANZA DEL FWE, PCE Y PFE. DEBE SER UN NUMERO ENTRE O  
C Y 100. POR EJEMPLO 95 REPRESENTA UN INTERVALO DE CONFIANZA  
C AL 95 PORCIENTO. NUMEROS NO APROPIADOS SE DESHECHAN  
C Y SE VUELVE A PREGUNTAR POR DICHO NIVEL. EL  
C INTERVALO DE CONFIANZA SIEMPRE SE OBTENDRA A DOS COLAS.  
C DEBIDO A QUE EL NUMERO DE SIMULACIONES DEBE SER 
C GRANDE(POR LO MENOS 40), SE USA LA APROXIMACION NORMAL. 
C EL PROGRAMA CALCULA EL VALOR DE Z(ALPHA) Y LO  
C ESCRIBE EN EL ARCHIVO DE SALIDA. 
C 
C   53 CONTINUE 
C      PRINT*,'  '                                                      
C      PRINT*,'   INTRODUCE EL NIVEL DE CONFIANZA' 
C      PRINT*,'   DEBE SER UN NUMERO ENTRE 0 Y 100' 
C      READ(*,*)RLEVEL 
C      IF(RLEVEL.LE.0..OR.RLEVEL.GE.100.)GO TO 53 
C      R1=RLEVEL/100.+(100.-RLEVEL)/200. 
C      ZT=ZQNT(R1)            

C 
C 
C EL NIVEL DE CONFIANZA Y LA Z DE TABLAS SE PONEN EN EL 
C ARCHIVO DE SALIDA. 
C       
C      WRITE(5,182)RLEVEL,ZT 
C  182 FORMAT(//10X,'NIVEL DE CONFIANZA: ',F5.2, 
C     1       //10X,'       Z DE TABLAS: ',F5.2)  
C 
C       
C EN ESTA PARTE SE PREGUNTA SI SE DESEA CORRER OTRA 
C SIMULACION. SI SE CONTESTA CON "1" ES "SI", 
C PARA EL "NO" PUEDE PONERSE CUALQUIER OTRO NUMERO.  
C SI SE METE UN CARACTER ALFABETICO,  
C EL PROGRAMA TERMINA CON UN MENSAJE DE ERROR. 
C       
  300 CONTINUE 
      PRINT*,'  '       
      PRINT*,'   DESEAS CORRER UNA SIMULACION?(SI=1)' 
      READ(*,*)IDESEA 
      IF(IDESEA.NE.1)GO TO 200 
C 
C 
C AQUI SE HACE LA LLAMADA A LA SUBRUTINA QUE HACE  
C LAS ESTIMACIONES DEL FWE, PCE Y PFE EN EL DISEÑO  
C COMPLETAMENTE AL AZAR. 
C 
C 
      CALL PCEPC(ISEMILLA) 
C 
C  
C SE REGRESA A PREGUNTAR SI SE DESEA ESTIMAR  
C EL FWE, PCE Y PFE EN OTRA COMBINACION DE  
C TRATAMIENTOS Y REPETICIONES. 
C  
      GO TO 300 
C 
C SI NO HAY OTRO EXPERIMENTO QUE ANALIZAR, LA  
C INSTRUCCIÓN DEL PRINCIPIO SALTARA AL 200 CONTINUE 
C QUE SE ENCUENTRA AQUI ABAJO. 
C       
  200 CONTINUE 
C 
C MENSAJE DE FIN DEL PROGRAMA AL ARCHIVO DE SALIDA. 
C   
      WRITE(5,184) 
C 
C SE CIERRA EL ARCHIVO DE SALIDA. 
C  
      CLOSE(5)   
C 
C UN MENSAJE FINAL DE FIN DEL PROGRAMA SE PONE EN  
C PANTALLA. 
C 
 
      WRITE(*,184) 
  184 FORMAT(///20X,'           FIN DEL PROGRAMA')     
C 
C 
C SE TERMINA LA EJECUCION. 



AANNEEXXOO  VV  
  

 
 
C       
      STOP 
      END 
C 
C       
      SUBROUTINE PCEPC(ISEMILLA) 
C 
C  
C SUBRUTINA CALCULAR EL VALOR P DE LAS COMPARACIONES 
C PARTICULARES EN LA PRUEBA D.E. EN UN 
C DISEÑO COMPLETAMENTE AL AZAR. 
C   
      dimension rmydata(20,20) 
      dimension RMED(20) 
      REAL CME,CMES 
      dimension rsim(20,20) 
      dimension RMEDsim(20) 
      DIMENSION NDERECH(20,20),RNDERECH(20,20) 
      DIMENSION VALSPC(20,20) 
      DIMENSION STAT(20,20),STATSIM(20,20) 
      DIMENSION NRECH(20,20) 
      DIMENSION VALSP(20,20) 
      CHARACTER*12 SALIDA 
      COMMON C0,C1,C2,D1,D2,D3 
      COMMON SALIDA 
      COMMON IB,IREPB 
C 
C 
C Se LEE la INFORMACION RELACIONADA CON EL NUMERO 
C DE TRATAMIENTOS Y REPETICIONES. 
C 
      PRINT*,'  ' 
      PRINT*,'   INTRODUCE EL NUMERO DE TRATAMIENTOS' 
      PRINT*,'   Y DE REPETICIONES'                             
      READ*,NTRAT,NREP 
C 
C 
C SE CALCULA EL NUMERO DE COMPARACIONES (N ESCOGE 2, ES 
C DECIR COMBINACIONES DE N TOMANDO 2 A LA VEZ). 
C 
      RNTRAT=NTRAT 
      RNCOMP=RNTRAT*(RNTRAT-1.)/2. 
      NCOMP=RNCOMP 
C 
C  
C EL NUMERO DE TRATAMIENTOS, REPETICIONES Y  
C COMPARACIONES SE MANDAN AL ARCHIVO DE SALIDA. 
C 
      WRITE(5,139)NTRAT,NREP,NCOMP 
  139 FORMAT(//10X,'NUMERO DE TRATAMIENTOS:',I4//     
     1         10X,'NUMERO DE REPETICIONES:',I4// 
     1         10X,'NUMERO DE COMPARACIONES:',I4) 
C 
C 
C LOS CONTADORES SE IGUALAN A CERO. 
C 
      DO 499 I=1,NTRAT-1 
      DO 499 J=I+1,NTRAT 
      NDERECH(I,J)=0 
  499 CONTINUE 

C 
C 
C EN ESTE LUGAR SE DA INICIO A LA SIMULACION. 
C IB=NUMERO DE SIMULACIONES. 
C 
                                  
      DO 322 NXX=1,IB 
C 
C 
C SE GENERAN DATOS DE UNA NORMAL CON MEDIA CERO Y  
C VARIANZA UNO. LOS DATOS SE GUARDAN EN LA MATRIZ 
C RMYDATA(20,20). ESTOS DATOS SON COMO DATOS  
C OBSERVADOS A LOS QUE SE LES CALCULA EL VALOR P. 
C 
      DO 883 I=1,NTRAT 
      DO 883 J=1,NREP 
      Y=UNIFO(ISEMILLA) 
      RMYDATA(I,J)=ZQNT(Y) 
  883 CONTINUE 
 
C 
C 
C SE INICIALIZA EL CONTADOR PARA EL NUMERO DE 
C RECHAZOS NRECH. ESTE ES EL CONTADOR QUE SIRVE  
C PARA CALCULAR EL PVALUE DE CADA COMPARACION  
C PARTICULAR EN LA PRUEBA DE. 
C 
      DO 333 I=1,NTRAT-1 
      DO 333 J=I+1,NTRAT 
      NRECH(I,J)=0 
  333 CONTINUE 
C 
C 
C SE LLAMA LA SUBRURINA PARA CALCULAR EL CUADRADO  
C MEDIO DEL ERROR Y LAS MEDIAS DE LOS TRATAMIENTOS 
C 
C. 
      CME=0.0 
      call resca1(ntrat,nrep,rmydata,RMED,cme) 
      write(*,*)'SIMULACION ',nxx 
C 
C SE ORDENAN LAS MEDIAS DE LOS TRATAMIENTOS. 
C                                            
      CALL SOLORD(RMED,NTRAT) 
C 
C SE CALCULAN LOS ESTADISTICOS. 
C      
      RNREP=NREP 
      DO 590 I=1,NTRAT-1 
      DO 590 J=I+1,NTRAT 
      STAT(I,J)=(RMED(J)-RMED(I))/SQRT(CME/RNREP) 
  590 CONTINUE 
C 
C AQUI EMPIEZAN LAS REPETICIONES POR SIMULACION 
C 
      do 321 nyy=1,IREPB 
C 
C 
C Se generan los datos normales 0,1. Los datos  
C se encuentran 

C en la matriz rsim. 
C 
      DO 83 I=1,NTRAT 
      DO 83 J=1,NREP 
      Y=UNIFO(ISEMILLA) 
      RSIM(I,J)=ZQNT(Y) 
   83 CONTINUE 
c 
c SE HACE EL ANALISIS DE VARIANZA. SE CALCULAN 
C CMES= CUADRADO MEDIO DEL ERROR SIMULADO 
c RSIM= MATRIZ 20x20 con datos simulados 
c RMEDSIM= MEDIAS de los datos simulados 
c                                          
      CMES=0.0 
      call resca1(ntrat,nrep,rsim,RMEDSIM,CMES) 
C 
C SE ORDENA EL VECTOR DE LAS MEDIAS DE LOS  
C TRATAMIENTOS SIMULADOS. 
C RMEDSIM=MEDIAS DE LOS TRATAMIENTOS. 
C NTRAT=NUMERO DE TRATAMIENTOS. 
C 
      call SOLORD(RMEDSIM,NTRAT) 
C 
C SE CALCULAN LOS ESTADISTICOS SIMULADOS. 
C                               
      RNREP=NREP 
      DO 858 I=1,NTRAT-1 
      DO 858 J=I+1,NTRAT 
      STATSIM(I,J)=(RMEDSIM(J)-RMEDSIM(I))/SQRT(CMES/RNREP) 
  858 CONTINUE 
C 
C 
C SE CUENTAN LAS VECES QUE HAY RECHAZOS. 
C  
      DO 859 I=1,NTRAT-1 
      DO 859 J=I+1,NTRAT 
      IF(STATSIM(I,J).GE.STAT(I,J))NRECH(I,J)=NRECH(I,J)+1 
  859 CONTINUE     
C 
C  
C AQUI SE TERMINA EL LOOP DE LAS ITERACIONES  
C POR SIMULACION 
C  
  321 CONTINUE 
C 
C 
C SE CALCULAN LOS VALORES P DE  LA  
C PRESENTE SIMULACION. 
C      
      RIREPB=IREPB 
      DO 860 I=1,NTRAT-1 
      DO 860 J=I+1,NTRAT 
      RNRECH=NRECH(I,J) 
      VALSP(I,J)=RNRECH/RIREPB 
  860 CONTINUE 
       
C 
C 
C LA PROPOSICION SIGUIENTE DA EL NUMERO DE  
C RECHAZOS POR POSICION. NOTESE QUE EL NIVEL DE 

C SIGNIFICANCIA SELECCIONADO ES 0.05. 
C 
      DO 861 I=1,NTRAT-1 
      DO 861 J=I+1,NTRAT 
      IF(VALSP(I,J).LE.0.05)NDERECH(I,J)=NDERECH(I,J)+1 
  861 CONTINUE 
C 
C 
C AQUI SE TERMINA EL LOOP DE LAS SIMULACIONES. 
C                                                          
  322 CONTINUE 
C 
C 
C EN ESTA PARTE SE CUANTIFICAN LOS ERRORES, Y SE  
C IMPRIMEN EN EL ARCHIVO DE SALIDA. 
C  
C     RECORDAR:   
C               RIB= NUMERO DE SIMULACIONES. 
C 
      RIB=IB 
      DO 855 I=1,NTRAT-1 
      DO 855 J=I+1,NTRAT 
      RNDERECH(I,J)=NDERECH(I,J) 
      VALSPC(I,J)=RNDERECH(I,J)/RIB 
      WRITE(5,863)I,J,VALSPC(I,J) 
  855 CONTINUE 
  863 FORMAT(//10X,' I= ',I2,' J= ',I2,' VALOR P= ',F8.5//    
     210X,'______________________________')    
 
C 
C SE REGRESA AL PROGRAMA PRINCIPAL. 
C  
      RETURN 
      END 
c 
c 
      subroutine resca1(ntrat,nrep,a,t,c) 
c 
C 
c Esta subrutina sirve para calcular LAS MEDIAS de 
c tratamientos y el cuadrado medio del error en un diseño 
c completamente al azar. 
C 
c Argumentos: NTRAT=NUMERO DE TRATAMIENTOS. 
c             NREP=NUMERO DE REPETICIONES. 
c             A=MATRIZ DE OBSERVACIONES. 
c 
c     Salida: 
c             t=VECTOR DE LAS MEDIAS DE LOS TRATAMIENTOS. 
c             c=cuadrado medio del error 
c   
      dimension a(20,20) 
      dimension t(20) 
C 
C INICIALIZACION DE LAS SUMAS NECESARIAS PARA 
C CALCULAR LAS MEDIAS Y EL CUADRADO MEDIO DEL ERROR. 
C       
      sum=0 
      sum2=0 
      sum2t=0.0 



AANNEEXXOO  VV  
  

 
 
C 
C LOS ENTEROS NREP Y NTRAT SE CONVIERTEN EN REALES 
C PARA CALCULOS POSTERIORES. 
C 
      RNREP=NREP             
      RNTRAT=NTRAT 
C 
C SUMA SOBRE TRATAMIENTOS E INICIALIZACION DEL 
C VECTOR DE MEDIAS T(20) 
C       
      do 2 i=1,ntrat 
      t(i)=0 
C 
C SUMA SOBRE LAS REPETICIONES 
C 
 
      do 3 j=1,nrep 
      sum=sum+a(i,j) 
      t(i)=t(i)+a(i,j) 
      sum2=sum2+a(i,j)*a(i,j) 
    3 continue 
      sum2t=sum2t+t(i)*t(i) 
      T(I)=T(I)/RNREP 
    2 continue 
C 
C SE CALCULA EL CUADRADO MEDIO DEL ERROR 
C 
      c=(sum2-sum2t/rnrep)/(rntrat*(rnrep-1.0)) 
      return 
      end 
C 
C 
      FUNCTION UNIFO(IX) 
C 
C 
C GENERADOR DE NUMEROS ALEATORIOS IMPLEMENTANDO LOS RECURSOS: 
C IX=16807*IX MOD (2**(31)-1) 
C USANDO SOLO 32 BITS, ICLUYENDO SIGNO. 
C 
C ALGUNOS COMPILADORES REQUIEREN LA DECLARACION: 
C INTEGER*4 IX, K1 
C  
C INPUT 
C IX= ENTERO MAYOR QUE 0 Y MENOR QUE 2147483647     
C 
C OUTPUT 
C IX= NUEVO VALOR PSEUDO-ALEATORIO, 
C UNIF= UNA FRACCION UNIFORME ENTRE 0 Y 1    
C 
C LA FUNCION GENERADORA DE NUMEROS ALEATORIOS 
C      
      K1=IX/127773 
      IX=16807*(IX-K1*127773)-K1*2836 
      IF(IX.LE.0.OR.IX.GE.2147483647)IX=IX+2147483647 
      UNIFO=IX*4.65661287e-10  
      RETURN 
      END 
C 
C           
      FUNCTION ZQNT(UNIF)  

C 
C 
C ESTA FUNCION EVALUA LOS CUANTILES 
C DE LA DISTRIBUCION NORMAL ESTANDAR 
C LAS CONSTANTES, C0,C1,C2 Y D1,D2,D3, DEBEN  
C SER DEFINIDAS EN EL PROGRAMA PRINCIPAL A SABER:  
C 
C      C0=2.515517 
C      C1=0.802853 
C      C2=0.010328 
C      D1=1.432788 
C      D2=0.189269 
C      D3=0.001308 
C 
C PARA MAYOR INFORMACION VER LA  
C FORMULA 26.1 DE: 
C 
C        ABAMOWITZ & STEGUN.  
C        A HANDBOOK OF MATHEMATICAL FUNCTIONS. 
C        NATIONAL BUREAU OF STANDARDS.  
C        WASHINGTON,DC. 
C     
      COMMON C0,C1,C2,D1,D2,D3 
      PP=UNIF 
      MULT=1 
      IF(PP.LT.0.5)MULT=-1 
      IF(PP.GT.0.5)PP=1-PP 
      if(pp.lt.1e-30)pp=1e-30 
      T=SQRT(LOG(1./(PP*PP))) 
      ZQNT=(T-((((C2*T+C1)*T)+C0)/(1+((((D3*T+D2)*T)+ 
     1D1)*T))))*MULT 
      IF(PP.EQ.0.5)ZQNT=0.0 
      RETURN 
      END 
C    
C 
      SUBROUTINE SOLORD(RMORD,NTRAT)     
C 
C     
C ESTA SUBRUTINA SIRVE PARA ORDENAR LAS MEDIAS  
C DE LOS TRATAMIENTOS EN FORMA ASCENDENTE.  
C LAS MEDIAS VIENEN EN EL VECTOR RMORD(20). LOS DATOS  
C QUEDARAN ORDENADOS EN EL MISMO VECTOR RMORD(20) 
C QUE SE RECIBIO COMO ARGUMENTO. 
C 
      DIMENSION RMORD(20)       
      DO 133 I=1,NTRAT-1 
      DO 133 J=I+1,NTRAT 
      IF(RMORD(I).LE.RMORD(J)) GO TO 133 
      T=RMORD(I) 
      RMORD(I)=RMORD(J) 
      RMORD(J)=T 
  133 CONTINUE 
      RETURN 
      END 

CORRIDA PARA COMPARACIONES PARTICULARES EN LA PRUEBA D.E 
NO PROTEGIDA 
 
 
 
          PROGRAMA PARA CALCULAR EL VALOR P DE LAS 
          COMPARACIONES PARTICULARES NO PROTEGIDAS DE   
          LA PRUEBA D.E. EN EL DISEÑO COMPLETAMENTE AL AZAR, 
 
          PROGRAMADO POR:PAOLA RUELAS VILLEGAS. 
                              ESTEBAN BURGUETE. 
                                JOSE TAMBORERO. 
 
 
 
 
          ARCHIVO DE  SALIDA: pdenp.lis     
 
 
          SEMILLA PARA SIMULACION:       98765 
 
 
          NUMERO TOTAL DE SIMULACIONES: 1000000 
          ITERACIONES  POR  SIMULACION:  10000 
 
 
 
 
          NUMERO DE TRATAMIENTOS:   5 
 
          NUMERO DE REPETICIONES:   4 
 
          NUMERO DE COMPARACIONES:  10 
 
 
           I=  1 J=  2 VALOR P=   .04860 
 
          ______________________________ 
 
 
           I=  1 J=  3 VALOR P=   .05100 
 
          ______________________________ 
 
 
           I=  1 J=  4 VALOR P=   .05230 
 
          ______________________________ 
 
 
           I=  1 J=  5 VALOR P=   .04920 
 
          ______________________________ 
 
 
           I=  2 J=  3 VALOR P=   .04630 
 
          ______________________________ 

           I=  2 J=  4 VALOR P=   .04800 
 
          ______________________________ 
 
 
           I=  2 J=  5 VALOR P=   .04950 
 
          ______________________________ 
 
 
           I=  3 J=  4 VALOR P=   .04880 
 
          ______________________________ 
 
 
           I=  3 J=  5 VALOR P=   .04640 
 
          ______________________________ 
 
 
           I=  4 J=  5 VALOR P=   .04890 
 
          ______________________________ 
 
 
 
                               FIN DEL PROGRAMA 
 
 
 


